

Maps, maps, and more maps using SAS[®] PROC SGMAP! (VR-045)

Melinda MacDougall is a Senior Data Programmer at Cincinnati Children's Hospital in the Division of General and Community Pediatrics. She has been using SAS[®] software for 15+ years as both a biostatistician and data programmer. She has worked on both NIH-funded clinical research studies as well as contracts for the Centers for Medicare & Medicaid Services.



Maps, maps, and more maps using SAS[®] PROC SGMAP! (VR-045)

Melinda MacDougall, MS
Cincinnati Children's Hospital Medical Center
Senior Data Programmer



Objectives

Showcase SAS mapping options at multiple levels

- National
- State
- Regional
- Local



Out with the Old, In with the New!

Old

- PROC GMAP
- MAPS/MAPSSAS libnames

New

- PROC SGMAP (SAS 9.4M5)
- MAPSGFK libname



National Maps United States



Simple Version

United States




Simple Version

```
proc sgmap mapdata=MAPSGFK.US noautolegend;  
    title "United States";  
    choromap / mapid=State;  
run;
```



Population (gradient)

United States - Population



Population (gradient)

```
proc sgmap mapdata=MAPSGFK.US maprespdata=US_pop  
noautolegend;  
  title "United States - Population";  
  choromap Population / mapid=State id=State  
    colormodel=(CXeff3ff CX08519c)  
    name='pop';  
  gradlegend 'pop' / title='Population'  
  extractscale;  
run;
```




Population (quintiles)

```
proc sgmap mapdata=MAPSGFK.US maprespdata=US_pop
noautolegend;
  title "United States - Population";
  choromap Pop_Quintile_c / name='pop'
  mapid=State id=State;
  styleattrs datacontrastcolors=(black)
             datacolors=(CXeff3ff CXbdd7e7 CX6baed6
                        CX3182bd CX08519c);
  keylegend 'pop' / title='Population';
run;
```

Population and Household Income

United States - Population and Household Income



Population and Household Income

```
proc sgmap mapdata=MAPSGFK.US maprespdata=US_pop  
plotdata=US_inc noautolegend;  
  title "United States - Population and Household Income";  
  choromap Pop_Quintile_c / name='pop' mapid=State id=State;  
  bubble x=x y=y size=HHIncome_median / name='income'  
    legendlabel='Relative Median Household Income'  
    fillattrs=(color=darkgray)  
    bradiusmin=1px bradiusmax=5px;  
  styleattrs datacontrastcolors=(black)  
    datacolors=(CXeff3ff CXbdd7e7 CX6baed6  
                CX3182bd CX08519c);  
  keylegend 'pop' / title='Population';  
  keylegend 'income' / autoitemsizes;  
run;
```



State Maps Ohio



Simple Version

Ohio



Simple Version


```
proc sgmap mapdata=Ohio_Counties1 noautolegend;  
  title "Ohio";  
  choromap / mapid=County;  
run;
```

¹ Base data set=MAPSGFK.US_Counties



Population (gradient)

Ohio - County Population




Population (gradient)

```
proc sgmap mapdata=Ohio counties  
  maprespdata=Ohio_pop noautolegend;  
  title "Ohio - County Population";  
  choromap Pop / mapid=County id=County  
    colormodel=(CXedf8e9 CX006d2c) name='pop';  
  gradlegend 'pop' / extractscale  
    title='Population';  
run;
```



Population and Household Income

Ohio - Population and Household Income



Median Household Income (county seats) ▲ < US median ▲ ≥ US median

Population and Household Income

```
proc sgmap mapdata=Ohio_counties plotdata=Ohio_inc  
maprespdata=Ohio_pop noautolegend;  
title "Ohio - Population and Household Income";  
choromap Pop / mapid=County id=County  
    colormodel=(CXedf8e9 CX006d2c) name='pop';  
scatter x=x y=y / name='income' group=HHIncome  
    markerattrs=(symbol=trianglefilled size=7);  
styleattrs datacontrastcolors=(black cxd01c8b);  
keylegend 'income' / autoitems  
    title="Median Household Income  
        (county seats)";  
run;
```




Regional Maps Greater Cincinnati



Simple Version

**Cincinnati Children's Hospital Medical Center
Primary Service Area**



Simple Version

```
proc sgmap mapdata=CCHMC1 noautolegend;  
  title "Cincinnati Children's Hospital Medical Center";  
  title2 "Primary Service Area";  
  choromap / mapid=County;  
run;
```

¹ Base data set=MAPSGFK.US_Counties




Add State Information

```
proc sgmap mapdata=CCHMC maprespdata=CCHMC noautolegend;  
  title "Cincinnati Children's Hospital Medical Center";  
  title2 "Primary Service Area";  
  choromap State / name='state' mapid=County id=County;  
  styleattrs datacontrastcolors=(black)  
             datacolors=(CXC6579A CX9AC449 CX9AD0DD);  
  keylegend 'state' / autoitems size title='State'  
            titleattrs=(weight=bold size=14pt)  
            valueattrs=(size=12pt);  
run;
```

Add County Names

Cincinnati Children's Hospital Medical Center Primary Service Area



State ■ Ohio ■ Kentucky ■ Indiana




Add County Names

```
proc sgmap mapdata=CCHMC maprespdata=CCHMC noautolegend;  
  title "Cincinnati Children's Hospital Medical Center";  
  title2 "Primary Service Area";  
  choromap State / name='state' mapid=County id=County;  
  styleattrs datacontrastcolors=(black)  
             datacolors=(CXC6579A CX9AC449 CX9AD0DD);  
  text x=x y=y text=CountyName / textattrs=(size=8pt)  
      position=center outline pad=0.5;  
  keylegend 'state' / autoitems size title='State'  
            titleattrs=(weight=bold size=14pt)  
            valueattrs=(size=12pt);  
  
run;
```



Add Clinic Locations

Cincinnati Children's Hospital Medical Center Primary Service Area



Clinic Type ★ Burnet Main Campus ☆ Liberty Campus
 △ Specialty Services + Primary Care Only



Add Clinic Locations

```
proc sgmap mapdata=CCHMC maprespdata=CCHMC
  plotdata=CCHMC_clinic noautolegend;
  title "Cincinnati Children's Hospital Medical Center";
  title2 "Primary Service Area";
  choromap State / name='state' mapid=County id=County;
  styleattrs datacontrastcolors=(black)
  datacolors=(CXC6579A CX9AC449 CX9AD0DD)
  datasymbols=(starfilled star triangle plus);
  scatter x=x y=y / name='type' markerattrs=(size=10);
  keylegend 'type' / autoitems size title='Clinic Type'
  titleattrs=(weight=bold size=12pt)
  valueattrs=(size=12pt);
run;
```




Local Maps

Hamilton County, Ohio



Simple Version

Hamilton County - Census Tracts



Simple Version


```
proc sgmap mapdata=tracts2 noautolegend;  
  title "Hamilton County - Census Tracts";  
  choromap / mapid=GEOID;  
run;
```

² Base data from census shapefiles: tl_2020_39_tract.shp



City of Cincinnati

Hamilton County - City of Cincinnati Census Tracts




City of Cincinnati

```
proc sgmap mapdata=tracts maprespdata=Cincy noautolegend;  
title "Hamilton County -  
      City of Cincinnati Census Tracts";  
choromap Cincinnati / mapid=GEOID id=GEOID  
      colormodel=(white CXde2d26);  
run;
```



City of Cincinnati

Hamilton County - City of Cincinnati Census Tracts



□ Outside city boundary □ Partially in city ■ Fully in city



City of Cincinnati

```
proc sgmap mapdata=tracts maprespdata=Cincy noautolegend;  
  title "Hamilton County -  
        City of Cincinnati Census Tracts";  
  choromap Type / mapid=GEOID id=GEOID discrete  
            name='Cincy';  
  styleattrs datacolors=(white cxfc9272 cxde2d26)  
            datacontrastcolors=(black);  
  keylegend 'Cincy' / autoitemsiz  
            valueattrs=(size=12pt);  
run;
```



Conclusion

PROC SGMAP is a powerful tool for making maps



References

- PROC SGMAP documentation:
(https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/grmapref/n18zwxj3couxmgn161trks2usk2v.htm)
- ColorBrewer 2.0 (<https://colorbrewer2.org/>)
 - Color palette options
- Census Data
 - Population and income data (<https://data.census.gov>)
 - Census tract shape files (<https://www.census.gov/cgi-bin/geo/shapefiles/index.php>)
- Plotting data on a Census block map [Robert Allison]
(<https://blogs.sas.com/content/graphicallyspeaking/2019/07/15/plotting-data-on-a-census-block-map/>)



Thank You!

- Melinda MacDougall, MS
- Cincinnati Children's Medical Center
- Senior Data Programmer
- Melinda.MacDougall@cchmc.org



Trademark Info

The output for this paper was generated using SAS software. Copyright © 2016 SAS Institute Inc. SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc., Cary, NC, USA.

